
Among the features added to recent AutoCAD® software releases, subdivision surfaces (also known as MESH entities) are especially interesting. These entities are exposed in the classic entity list format. So, our good old AutoLISP® can be used for creating and modifying them. By calculating the coordinates for their vertices from different math formulas, we can create an unlimited variety of 3D shapes. By smoothing these meshes, converting them to surfaces and solids, and applying the different surface and solid editing tools, we can explore a vast universe of 3D forms. These forms can be used in other Autodesk® software like Revit or MAX for the development of new design concepts.
Learning Objectives

At the end of this class, you will be able to:

· Understand, load, test and execute AutoLISP code.

· Automate 3D Point of View and Visual Styles.

· Understand the structure of MESH DXF group codes.
· Create and modify MESH entities through AutoLISP programming.
· Understand and apply different AutoCAD 3D object conversion an edition commands.
About the Speaker
Reinaldo N. Togores has been using AutoCAD for about 25 years in his work as an Architect and Industrial Designer. For the past eighteen years he has taught AutoLISP programming to Civil and Industrial Engineering Students at the University of Cantabria, in Northern Spain. As a researcher on Computer Aided Design topics he has worked with several research groups at the University of Cantabria and has been a member of the Autodesk Developers Network since the late ‘90s. He has authored two books about Visual LISP programming (2003 and 2012).

 http://www.togores.net/
 togoresr@unican.es
togores.design@gmail.com
Understand, load, test and execute AutoLISP code
The Visual LISP IDE
The Visual LISP Integrated Development Environment (VLISP IDE) includes tools that assist in writing, modifying and debugging the source code. Now we can, once we have coded the program using the Programming Editor, check its operation step by step, viewing at all times the output returned by the evaluation of each one of the expressions and the value the variables are assuming without having to include any additional expressions in the program. It also provides the means to package and distribute applications in a single executable file that contains the DCL dialog interface. These operations are performed within a single environment, allowing text editing and program debugging, plus interaction with AutoCAD and other applications.

Visual LISP has its own set of windows and menus, but cannot run independently of AutoCAD. A working session is started by typing vlide or vlisp on the command line.
Interactivity: The Visual LISP Console. 

From the Console, you can run LISP expressions to check their outcome. This can also be done from the AutoCAD command line but there are several differences in how the two operate. For example, to display the value of a variable in LISP, we simply type the name of the variable in the Console window and click ENTER. But to obtain the value of a variable in the AutoCAD command line, we must precede the variable name with an exclamation point (!).
The main function of the console is to evaluate LISP expressions. The symbol _$ indicates the Console is waiting for an expression to be evaluated. The expression can be typed following this symbol. The text you type into the console is not evaluated until you explicitly press ENTER. To continue a LISP expression on a new line avoiding its evaluation you can use the key combination CTRL+ENTER.
The Programming Editor: 



It is designed especially for use with AutoLISP code. For each source code file a new Editor window opens with the name of that file in its title bar. The name also appears in the status bar. An asterisk (*) next to the file name in the status bar indicates that you have made changes that have not been saved. The Editor as well as the Console, is designed to assist in AutoLISP programming and has the ability to complete expressions, locate the matching opening and closing parentheses, syntax coloring and evaluating expressions without leaving the Editor window. In contrast to the Console, pressing ENTER generates a new line in the Editor. The new line is automatically indented. The Editor is also able to format the code according to different user selectable styles. In general, besides syntax color coding and automatic indentation, the Editor provides other utilities that assist in program development. Some of these utilities are:

· Checking the parenthesis closure.

· The ability to insert expression-closing comments. 

· Commenting and uncommenting text blocks. 

· Finding and replacing text. 

· Syntax checking. 

· Loading of LISP expressions to be evaluated in the Console.

The Visual LISP code formatter organizes the content of the Editor's active window to improve its appearance and readability. Although intelligent indentation organizes the program code as the text is typed, it can also be explicitly invoked with the Tools toolbar buttons to rearrange the selected blocks or the entire Editor text. 
To start programming in Visual LISP two are the tools we must master: the Console and Editor. The simultaneous use of both, Console and Editor, is one of the keys to success in Visual LISP programming.

· Working in the Visual LISP Console we can understand why we say that LISP is an interactive language. If we type an expression in the Console, we'll obtain its value immediately. This way we can check the expressions that afterwards we will include in our programs.

· It is in the Editor where we establish the final form of our programs, saving them to the data storage units of our system.

Automate 3D Point of View and Visual Styles
The representations of a 3D object on the screen or in print, in both cases projected onto a 2D plane, must include the information necessary for the observer to identify, for a particular orientation in space, which parts of the object are closer to the observer. These techniques include parallel and perspective projections, especially when combined with techniques for differentiating the representation of visible and hidden edges or directly removing hidden lines and shading faces according to their angle relative to the viewing direction or to the position of a light source. To automate this process so that we can immediately appreciate the resulting volumes, we have developed functions (Listings 1 and 2) that can be added to any 3D modeling program.
Visual appearance.

Besides the Viewpoint orientation, the perception of 3D forms is enhanced by the color, the shading and the lighting of objects. AutoCAD provides a set of predefined visual styles that can be set using the ‑VISUALSTYLES command. These predefined visual styles are:

"_W" = Wireframe; "_H" Hidden lines; "_R" Realist; "_C" Conceptual; "_S" Shaded; "_E" Shaded with Edges; "_G" Shades of Gray; "_SK" Sketchy; "_X" X-ray. The appearance of these styles is controlled from a number of system variables that make it possible to program custom visual styles. These variables are described in Table 1.

Table 1. Variables that control the visual appearance.

	Variable
	Description
	Values

	VSBACKGROUNDS
	Controls whether backgrounds are displayed. 
	0: Off / 1: On

	VSEDGECOLOR
	Color for edges.
	Color index or RGB.

	VSEDGEJITTER
	Edges of 3D objects appear wavy.
	1: Low / 2: Medium /
3: High

	VSEDGELEX
	Extends edges beyond their intersection.
	From 1 to 100 pixels.

	VSEDGEOVERHANG
	Extends edges beyond their intersection.
	From 1 to 100 pixels.

	VSEDGES
	Controls the types of edge that are displayed.
	0: None / 1: Isolines /
2: Face edges

	VSEDGESMOOTH
	Angle at which crease edges are displayed.
	Between 0 and 180.

	VSFACECOLORMODE
	Controls how the color of faces is calculated.
	0: Normal / 1: Monochrome / 2: Tint /
3: Desaturate

	VSFACEHIGHLIGHT
	Controls the display of specular highlights on faces without materials.
	Between -100 and 100.

	VSFACEOPACITY
	Transparency level for 3D objects.
	Between 100% and 0% opacity.

	VSFACESTYLE
	Faces display mode.
	0: No / 1: Real / 2: Gooch

	VSHIDEPRECISION
	Precision for hides and shades.
	0: Simple / 1: Double

	VSINTERSECTIONCOLOR
	Color of intersection polylines.
	Color index or RGB.

	VSINTERSECTIONEDGES
	Display of intersection edges.
	0: Off / 1: On

	VSINTERSECTIONLTYPE
	Intersections Linetype.
	Between 0 and 11.

	VSISOONTOP
	Displays Isolines on top of shaded objects.
	0: Off / 1: On

	VSLIGHTINGQUALITY
	Lighting quality.
	0: Faceted / 1: Smooth /
2: Smoothest

	VSMATERIALMODE
	Display of materials.
	0: None / 1: Materials /
2: Materials and Textures

	VSMONOCOLOR
	Color for monochrome and tint display of faces.
	RGB color.

	VSOBSCUREDCOLOR
	Color of hidden lines.
	Color index or RGB.

	VSOCCLUDEDEDGES
	Controls whether hidden edges are displayed.
	0: Off / 1: On

	VSOCCLUDEDLTYPE
	Linetype of hidden lines.
	Between 0 and 11.

	VSSHADOWS
	Controls whether to show shadows.
	0: No Shadows / 1: Ground shadows / 2: Full shadows

	VSSILHEDGES
	Controls the display of silhouette edges.
	0: Off / 1: On

	VSSILHWIDTH
	Width in pixels of the silhouette edges.
	Between 1 and 25.

	PERSPECTIVE
	Specifies whether the view is displayed in perspective.
	0: Off / 1: On


To set a visual style for our programs we have defined the var-vis function (Listing 1) which customizes the way 3D objects are presented. The effects of the proposed settings can be seen on this book's cover. We begin by setting a color for the VSMONOCOLOR variable so that there is no need for setting the Layer's or the object's color, as it is linked to the style set for the viewport. The variable's value is set to a RGB color with the string "RGB: 211,76,3" as the to obtain the shade of orange we wanted. But for this color to be applied to our 3D objects' surface the variable VSFACECOLORMODE must be set to 1 instead of its default value 0.

To enhance details a Gooch shading is used instead of a realistic one. Gooch shading uses warm and cool tones instead of dark and light, as a means to display shadows. This mode is enabled by setting the variable VSFACESTYLE as 2. For a realistic presentation its value should be set to 1.

In this view Isolines are displayed in a lighter color than the surfaces as a way to emphasize the model's relief. To attain this it has been necessary to ensure that the VSEDGES variable's value is 1 so that the Isolines are displayed and to set their color in the VSEDGECOLOR variable as "RGB 255,212,82". Finally, the VSISOONTOP variable has been set to 0. This way the Isolines in the back part of the model will remain hidden.

The visibility of the obscured (hidden) edges is controlled by the VSOBSCUREDEDGES variable. Setting its value to 0 will not display the hidden edges. Another type of edges is the one corresponding to intersections. These are the edges generated when where two different objects intersect and their visibility is enabled or disabled by the VSOCCLUDEDEDGES system variable. For the cover image, being a single object, there will be no difference setting its value On (1) or Off (0).
(defun var-vis  ()

  (setvar "VSMONOCOLOR" " RGB:211,76,3")

  (setvar "VSFACECOLORMODE" 1)

  (setvar "VSEDGES" 1)

  (setvar "VSEDGECOLOR" "RGB:255,212,82")

  (setvar "VSISOONTOP" 0)

  (setvar "VSFACESTYLE" 2)

  (setvar "VSOBSCUREDEDGES" 0)

  (setvar "VSOCCLUDEDEDGES" 0)

  (setvar "PERSPECTIVE" 1))

Listing 1. Function that sets a custom visual style.
Setting a Southwest isometric point of view.
The point of view is set using the DVIEW command. This is our first approximation to the automation of commands through AutoLISP. Listing 2 shows the function definition. The arguments passed to the DVIEW command are a double quote as the answer to the Select objects or <use DVIEWBLOCK>: prompt, “_ca” for selecting the Camera option, 45 as the answer to the Specify camera location, or enter angle from XY plane and -135 as the answer to the Specify camera location, or enter angle in XY plane from X axis. A double quote is necessary for ending the command. A ZOOM EXTENSION is then executed and finally a call to the var-vis function we defined in Listing 1.

 (defun cmd-SW ()

  (command "_dview" "" "_ca" 45 -135 "")

  (command "_zoom" "_e")

  (var-vis))
Listing 2. Setting the SW isometric view and Visual Style.

Understand the structure of MESH DXF group codes
A MESH is a surface object. That is, unlike 3D SOLIDS it has no volume or mass properties. It is made up of faces which in turn are surrounded by edges and these edges are specified by pairs of vertices. MESHES belong to the kind of objects known in Computer Graphics as Subdivision Surfaces. These objects can have different SMOOTHING levels. Increasing the smoothing level generates facets that increasingly approximate rounded forms.
We have two ways of making MESH objects through AutoLISP. We can use it as a scripting language to automate the command interface. Or we can use the AutoLISP ENTMAKE function to create MESH objects directly without the limitations of the different commands. This requires a thorough understanding of the way its properties are exposed through DXF Group Codes.

MESH entities are defined from Faces, Edges and Vertices. Faces are defined from the sequence of vertices that surround them. The sequence of two vertices defines an edge and face adjacency is determined by their shared edges. This basic MESH structure, its faces, edges and vertices, are designated as a Level 0. 

To explain how the MESH entity definition list is structured we shall use as an example a tetrahedral mesh created using one of this class’ sample programs. The entity list obtained with entget is discussed in detail in Table 2.
Table 2. MESH entity definition list.

	((-1 . <Entity name: 7ffff605b00>)

 (0 . "MESH")   ; Entity type
 (330 . <Entity name: 7ffff6039f0>)

 (5 . "228")

 (100 . "AcDbEntity")  ; Subclass

 (67 . 0)

 (410 . "Model")

 (8 . "0")

 (100 . "AcDbSubDMesh"); Subclass
	Common group codes.

Group codes -1, 330 and 5 are specific to each object and are not included in the list passed to entmake.

Group codes 67, 410 and 8 corresponding to ModelSpace/Paper, Layout and Layer can be omitted. Current values are automatically assigned.

	  (71 . 2) ; Version number

 (72 . 0) ; Blend Crease property

 (91 . 0) ; Subdivision level


	Version and smoothing.

Version number is always 2. Value 1 is admitted. Any other will cause an error.

Blend Crease determines whether the crease will be smoothed according to the ZOOM value. This does not produce any effect.

Subdivision level number determines the object's smoothness.

	 (92 . 4); Num. vertices Level 0

 (10 0.0 0.0 100.0)    ; Vert. 0

 (10 0.0 94.2 -33.3)   ; Vert. 1

 (10 -81.6 -47.1 -33.3); Vert. 2

 (10 81.6 -47.1 -33.3) ; Vert. 3
	Vertices position. 

Group code 92 indicates the number of vertices. Must be followed by as many group codes 10 as vertices. Each group code 10 sub-list holds a vertex's coordinates.

	  (93 . 16);Level 0 facelist #
  (90 . 3) ; # vertices Face 1

  (90 . 0)

  (90 . 1)

  (90 . 2)

  (90 . 3) ; # vertices Face 2

  (90 . 0)

  (90 . 2)

  (90 . 3)

  (90 . 3) ; # vertices Face 3

  (90 . 0)

  (90 . 3)

  (90 . 1)

  (90 . 3) ; # vertices Face 4

  (90 . 1)

  (90 . 3)

  (90 . 2)
	Faces Definition.

Group code 93 indicates the number of sublists included in Level 0 faces definition. Each face is defined from a block of sublists associated to group code 90, the first of which holds the number of vertices around the face and is followed by as many group code 90 sublists as vertices. That is, each block consists of number-of-vertices  + 1 sublists.

The associated value for each sublist is the index (zero-based) identifying the vertex in the group code 10 sequence.

Unlike what happens with the PolyfaceMesh, the number of vertices is not limited to 4.

	  (94 . 6) ; Edge count level 0

  (90 . 0) ; Edge 1

  (90 . 1)

  (90 . 0) ; Edge 2

  (90 . 2)

  (90 . 0) ; Edge 3

  (90 . 2)

  (90 . 0) ; Edge 4

  (90 . 3)

  (90 . 0) ; Edge 5

  (90 . 3)

  (90 . 0) ; Edge 6

  (90 . 3)
	Edges definition.

Group code 94 indicates the number of edges for Level 0. 

The sublists that follow associated with group code 90 double that number, since each edge is defined by two successive sublists. For example, Edge 1 is in this case the one that connects vertices 0 and 1.

	  (95 . 6)    ; # of creases

  (140 . 0.0) ; Edge 1

  (140 . 0.0) ; Edge 2

  (140 . 0.0) ; Edge 3

  (140 . 0.0) ; Edge 4

  (140 . 0.0) ; Edge 5

  (140 . 0.0) ; Edge 6


	Edge creases.

Group code 95 matches group code 94 as it counts the number of edges. For each edge, group code 140 indicates the highest level of smoothing for which the crease is maintained. If the level of smoothing is increased, this crease is also smoothed. A value of -1 indicates that the crease is always maintained. A value of 0 indicates there is no crease.

	  (90 . 1) ; Modified subentities

  (91 . 9) ; Subentity ID

  (92 . 1) ; Number of properties

  (90 . 0) ; Property (0=Color)

  (63 . 1) ; Value (1=Red)


	Properties modification.

The group code 90 that follows the 140 group codes indicates the number of sub-entities whose properties have been modified. Group code 91 indicates the modified subentity. The sub-entity numbers are zero-based, starting with the edges. In this case, the index 9 sub-entity is the last face (0 to 5 for edges + 4 faces). The group code 90 following the 92 group code indicates the modified property: 0 = Color 1 = Material, 2 = Transparency, 3 = Material mapping. In this case the color has been changed to red (code 63).


Create and modify MESH entities through AutoLISP programming
Commands can be used to create MESH objects. The MESH command supplies a set of so called “primitives”, an assortment of simple shapes like boxes, spheres, cylinders, and so on. The number of faces (or tesselations) for MESH primitives can be set using the _MESHOPTIONS command, but this command uses a dialog box that cannot be suppressed, so it is not adequate for its use in AutoLISP programs. In AutoLISP programs we’ll rather use a set of system variables identified by the DIVMESH prefix that set the tessellations for each of the primitive’s dimensions. These variables are described in Table 3.

Table 3. System variables that control MESH subdivisions.

	Mesh:
	Variable:
	Subdivisions:

	_Box
	DIVMESHBOXHEIGHT
	Along the Z axis. Values between 1 and 256.

	
	DIVMESHBOXLENGTH
	Along the X axis. Values between 1 and 256.

	
	DIVMESHBOXWIDTH
	Along the Y axis. Values between 1 and 256.

	_Cone
	DIVMESHCONEAXIS
	Along the base perimeter. Values between 3 and 256.

	
	DIVMESHCONEBASE
	Along the radius. Values between 1 and 256.

	
	DIVMESHCONEHEIGHT
	Along its height. Values between 1 and 256.

	_CYlinder
	DIVMESHCYLAXIS
	Along the base perimeter. Values between 3 and 256.

	
	DIVMESHCYLBASE
	Along the radius. Values between 1 and 256.

	
	DIVMESHCYLHEIGHT
	Along its height. Values between 1 and 256.

	Pyramid
	DIVMESHPYRBASE
	Along the base radius. Values between 1 and 256.

	
	DIVMESHPYRHEIGHT
	Along its height. Values between 1 and 256.

	
	DIVMESHPYRLENGTH
	Along each side of the base. . Values between 1 and 256.

	_Sphere
	DIVMESHSPHEREAXIS
	Along its equator Values between 3 and 256.

	
	DIVMESHSPHEREHEIGHT
	Along the Z axis. Values between 2 and 1024.

	_Torus
	DIVMESHTORUSPATH
	Along the sweep path. Values between 3 and 512.

	
	DIVMESHTORUSSECTION
	Along the profile's circumference. Between 3 and 512.

	_Wedge
	DIVMESHWEDGEBASE
	From the triangular face's centroid to its sides. 1 to 64.

	
	DIVMESHWEDGEHEIGHT
	Along the Z axis. Values between 1 and 64.

	
	DIVMESHWEDGELENGTH
	Along the X axis. Values between 1 and 64.

	
	DIVMESHWEDGESLOPE
	Along the inclined face. Values between 1 and 64.

	
	DIVMESHWEDGEWIDTH
	Along the Y axis. Values between 1 and 64.


MESH objects are now also created by the old PolygonMesh creating commands, RULESURF, TABSURF, REVSURF and EDGESURF. In these cases the number of faces depends on the SURFTAB1 and SURFTAB2 system variables. 
Creating a Rectangular Prism MESH.

To aid in the creation of meshes using these commands we can define functions that automate the process of setting the subdivision variables. Listing 3 shows a function that creates a rectangular prism.
 (defun cmd-mesh-box

       (center dim-x dim-y dim-z div-x div-y div-z /)

  (setvar "DIVMESHBOXLENGTH" div-x)

  (setvar "DIVMESHBOXWIDTH" div-y)

  (setvar "DIVMESHBOXHEIGHT" div-z)

  (vl-cmdf "._MESH" "_B" "_C" center "_L" dim-x dim-y dim-z))

Listing 3. Creating a mesh BOX using the MESH command.

The CMD-MESH-BOX function takes as arguments a point to be used as the box's centroid, the X, Y and Z dimensions, and the number of length, width and height subdivisions. In this case the old AutoLISP command function is replaced by the newer vl-cmdf function. The options used for the MESH command are “_B” for the BOX primitive type, “_C” for the CENTER option and “_L” for LENGTH.
Creating a flat square mesh

Listing 4 shows a function that creates a flat square MESH using the EDGESURF command.  

 (defun cmd-square-mesh  (ins-pt side div / x-min x-max y-min y-max z edges)

  (setvar "MESHTYPE" 1)                                   ; MESH type object

  (setvar "SURFTAB1" div)                                 ; U tesselations

  (setvar "SURFTAB2" div)                                 ; V tesselations

  (setq x-min (nth 0 ins-pt)

        x-max (+ x-min side)

        y-min (nth 1 ins-pt) 

        y-max (+ y-min side)

        z     (nth 2 ins-pt))

  (vl-cmdf "_LINE" ins-pt (list x-max y-min z) "")               ; Line 1

  (setq edges (cons (entlast) edges)) 

  (vl-cmdf "_LINE" (list x-max y-min z) (list x-max y-max z) "") ; Line 2

  (setq edges (cons (entlast) edges))

  (vl-cmdf "_LINE" (list x-max y-max z) (list x-min y-max z) "") ; Line 3

  (setq edges (cons (entlast) edges))

  (vl-cmdf "_LINE" (list x-min y-max z) ins-pt "")               ; Line 4

  (setq edges (cons (entlast) edges))

  (apply 'vl-cmdf (cons "EDGESURF" edges))                ; Create MESH

  (mapcar 'entdel edges)                                  ; Erase lines

  (princ))

Listing 4. Creating a planar square MESH using the EDGESURF command.

The CMD-SQUARE-MESH function receives as arguments a point (ins-pt) which will be the lower left corner of the mesh, le length of the square’s side and the number of divisions which will be the same along the X and Y directions. As the lines for the edges are created their ename (obtained through entlast) is added to the EDGES list. This list will be used for supplying the lines to EDGESURF and to erase them once the MESH has been created. As the edges are contained in a lists two rather sophisticated constructs are used. To create the MESH the APPLY function is used which passes a list of arguments to, and executes, a specified function: (apply 'vl-cmdf (cons "EDGESURF" edges)) 
This way the string “EDGESURF” is added as the first term to the list that already contains the lines’ enames and ‘VL-CMDF is applied to this list. To erase the lines the MAPCAR function that executes a function with a list supplied as argument is used.

Creating MESH objects through ENTMAKE.
To create MESH entities without using AutoCAD commands we propose the ENT-MESH function (Listing 5). ENT-MESH receives three list arguments, VERTICES, FACES and EDGES.

· The VERTICES list will contain the coordinates for each Vertex. 
· The FACES list contains sublists, each of them indicating the vertices that surround a face. The vertices are specified by indices that refer to the VERTICES list. 
· The EDGES list will hold a sequence of pairs of vertex indices. Each pair indicates the starting and ending point of an Edge.
The ENT-MESH function begins by creating the HEADER list with fixed values. The smoothing level (group code 91) is set to ZERO. 
From the information contained in the VERTICES, FACES and EDGES lists ENT-MESH creates the entity definition sub-lists, associated with the corresponding DXF group codes. The process is:

1. Find the length of each list (VERTICES, FACES and EDGES) and associate it with the group code (92 for vertices, 93 for faces and 94 for edges) that indicates the number of items. 
2. Add each item's sublist.
The items contained in the VERTICES are added using the CONS function in a FOREACH loop to a list assigned to the variable ENT-LIST. 
Adding the FACES data is somewhat more involved. It’s done in two nested FOREACH loops. The value associated to group code 93 is the sum of the number of sublists in FACES which is returned by the (length FACES) expression, plus the total number of vertices that surround the faces which is found through the expression (length (apply ‘append FACES)). This works by:
3. Flattening the list with (apply ‘append FACES)

4. Finding the list of the “flattened” list using the LENGTH function.
The expression which does this is:

(setq

    ent-list (cons (cons 93

                         (+ (length faces)

                            (length (apply 'append faces))))

                   ent-list))
Once the number to be assigned to group code 93 is found it will be necessary to add the code 90 values. For each face we must add:

5. A group code 90 that indicates the number of vertices surrounding that face.

6. As many group code 90 values as vertices indexes contained in that face sublist.

The following nested FOREACH expression is used to manage this:

(foreach face  faces

    (setq ent-list (cons (cons 90 (length face)) ent-list))

    (foreach vertexindex  face

      (setq ent-list (cons (cons 90 vertexindex) ent-list)))
As each edge is defined in the EDGES list by pairs of vertex indices, the number of edges to be used as the value for group code 94 will be half the length of the EDGES list. The values for the group code 90 sublists that follow will be assigned in a single FOREACH loop just as we did for the VERTICES data.
Although no information will be added for creases, the number of creases must be added to the header. The number of creases is the same as the number of edges. The header is then appended to the reversed ENT-LIST and ENTMAKE is then called with this final ENT-LIST as argument.
(defun ent-mesh  (vertices faces edges / header ent-list)

  ;;Entity header:

  (setq header '((0 . "MESH")

                 (100 . "AcDbEntity")

                 (100 . "AcDbSubDMesh")

                 (71 . 2)

                 (72 . 0)

                 (91 . 0)))

  ;;Vertices data:

  (setq ent-list (cons (cons 92 (length vertices)) ent-list))

  (foreach vertex  vertices

    (setq ent-list (cons (cons 10 vertex) ent-list)))

  ;;Faces data:

  (setq

    ent-list (cons (cons 93

                         (+ (length faces)

                            (length (apply 'append faces))))

                   ent-list))

  (foreach face  faces

    (setq ent-list (cons (cons 90 (length face)) ent-list))

    (foreach vertexindex  face

      (setq ent-list (cons (cons 90 vertexindex) ent-list))))

  ;;Edges data:

  (setq ent-list (cons (cons 94 (/ (length edges) 2)) ent-list))

  (foreach vertexindex  edges

    (setq ent-list (cons (cons 90 vertexindex) ent-list)))

  ;;Crease data:

  (setq ent-list (cons (cons 95 (/ (length edges) 2)) ent-list))

  (setq ent-list (append header (reverse ent-list)))

  (entmake ent-list))

Listing 5. Creating a MESH through ENTMAKE.
AutoCAD doesn’t have Tetrahedron or Dodecahedron MESH primitives. Using the ENT-MESH function we can define two new commands, TETRAHEDRON (Listing 6) and DODECAHEDRON (Listing 7) that will draw these polyhedrons with a user-defined radius. It will be centered on the coordinate system’s origin to make it simple. In my book it is shown how to place it in anywhere and aligned to any user defined coordinate system. These functions are named with the C: prefix so they can be called from the AutoCAD command line without the need for parenthesis, just like any other native command. Both these commands prompt for the Radius, set the values for the VERTICES, FACES and EDGES list variables and then call ENT-MESH. The resulting MESH is then scaled using the SCALE command. Finally the point of view and visual style is set by calling CMD-SW.
(defun C:TETRAHEDRON  (/ radius vertices faces edges)

  (setq radius   (getdist '(0 0 0) "\nCircumscribed sphere radius: ")

        vertices '((0 0 1)

                   (0 0.94280904 -0.33333333)

                   (-0.81649658 -0.47140452 -0.33333333)

                   (0.81649658 -0.47140452 -0.33333333))

        faces    '((0 1 2) (0 2 3) (0 3 1) (1 3 2))

        edges    '(0 1 1 2 2 0 2 3 3 0 3 1))

  (ent-mesh vertices faces edges)

  (command "_scale" (entlast) "" '(0 0 0) radius)

  (cmd-SW))

Listing 6. TETRAHEDRON command.
(defun C:DODECAHEDRON  (/ radius vertices faces edges)

  (setq radius   (getdist '(0 0 0) "\nCircumscribed sphere radius: ")

        vertices '((0.57735 -0.187592 0.794654)

                   (0.356822 0.491123 0.794654)

                   (-0.356822 0.491123 0.794654)

                   (-0.57735 -0.187592 0.794654)

                   (-1.11022e-016 -0.607062 0.794654)

                   (0.934172 -0.303531 0.187592)

                   (0.934172 0.303531 -0.187592)

                   (0.57735 0.794654 0.187592)

                   (-5.55112e-017 0.982247 -0.187592)

                   (-0.57735 0.794654 0.187592)

                   (-0.934172 0.303531 -0.187592)

                   (-0.934172 -0.303531 0.187592)

                   (-0.57735 -0.794654 -0.187592)

                   (5.55112e-017 -0.982247 0.187592)

                   (0.57735 -0.794654 -0.187592)

                   (0.356822 -0.491123 -0.794654)

                   (0.57735 0.187592 -0.794654)

                   (-1.33234e-016 0.607062 -0.794654)

                   (-0.57735 0.187592 -0.794654)

                   (-0.356822 -0.491123 -0.794654))

        faces    '((0 1 2 3 4)

                   (0 5 6 7 1)

                   (1 7 8 9 2)

                   (2 9 10 11 3)

                   (3 11 12 13 4)

                   (4 13 14 5 0)

                   (5 14 15 16 6)

                   (7 6 16 17 8)

                   (9 8 17 18 10)

                   (11 10 18 19 12)

                   (13 12 19 15 14)

                   (15 19 18 17 16))

        edges    '(0  1  1  2  2  3  3  4  4  0  0  5  5  6  6  7

                   7  1  7  8  8  9  9  2  9  10 10 11 11 3  11 12

                   12 13 13 4  13 14 14 5  14 15 15 16 16 6  16 17

                   17 8  17 18 18 10 18 19 19 12 19 15))

  (ent-mesh vertices faces edges)

  (command "_scale" (entlast) "" '(0 0 0) radius)

  (cmd-SW))
Listing 7. DODECAHEDRON command.
Ways to MOD the MESH: ENTMOD
ENTMOD receives an entity’s modified definition list and transforms it as specified by the new DXF group code values. The list is usually modified using the SUBST function. To modify the shape of a MESH we will change its vertex coordinate values which are associated to group code 10. Being associated to the same DXF group code, we need a function that will collect all these values. The auxiliary function VALUES (Listing 8) receives a Group Code as the KEY argument and an entity definition as the LST argument, returning a list with all the values associated to the KEY Group Code.
 (defun values  (key lst / sublist result)

  (while (setq sublist (assoc key lst))

    (setq result (cons (cdr sublist) result)

          lst    (cdr (member sublist lst))))

  (reverse result))

Listing 8. Extraction of multiple values from an association list.

To automate the transformation every vertex’s Z coordinate we have defined the ENTMOD-Z function (Listing 9). This function receives a MESH’s ename and a function that receiving the X and Y values will calculate and return a new Z value. ENTMOD-Z will obtain the MESH entity list, assigning it to the ENT-LIST variable and applying the VALUES function it will extract a list with all of the vertices coordinates, assigning them to the VERTICES variable. 

The VERTICES list is then processed in a FOREACH loop, extracting the X and Y values of each vertex and calculating the new Z coordinate by applying the EQUATION function received as argument to the X and Y vertices.
Once the value of Z is computed the XYZ vertex coordinates list is added to the VERTICES-MOD list using the CONS function. When the FOREACH loop is concluded, the VERTICES-MOD list is reversed to regain the original vertices order. 
The entity definition list will be modified in successive calls to ENTMOD. This is done in a WHILE loop which traverses in parallel both the original VERTICES and the modified VERTICES-MOD lists. When the loop concludes all the vertices will have been modified in the entity definition list assigned to the ENT-LIST variable. Calling ENTMOD with ENT-LIST as argument will apply these modifications to the original MESH. 
AU-2012-MESH lisp code includes ten functions (named S01 to S10) which can be used to calculate a value as a function of two variables. When passing one of these function names to ENTMOD-Z the function name must be preceded by an apostrophe symbol. 
(defun entmod-Z

       (ename equation / ent-list vertices x y z vertices-mod)

  (setq ent-list (entget ename)

        vertices (values 10 ent-list))

  (foreach vertex  vertices

    (setq x            (nth 0 vertex)

          y            (nth 1 vertex)

          z            (apply equation (list x y))

          vertices-mod (cons (list x y z) vertices-mod)))

  (setq vertices-mod (reverse vertices-mod))

  (setq i 0)

  (while (< i (length vertices))

    (setq ent-list (subst (cons 10 (nth i vertices-mod))

                          (cons 10 (nth i vertices))

                          ent-list)

          i        (1+ i)))

  (entmod ent-list))
Listing 9. Modifying vertex Z coordinates through entmod
Ways to MOD the MESH: ENTMOD
ActiveX objects expose Properties whose values can be read and sometimes modified. Modifying these Properties has the same effect we have already obtained through ENTMOD. It can be faster and require less programming effort but they are not available in AutoCAD for Mac.
To replicate the functionality of our ENTMOD-Z function we propose the AX-MOD-Z function (Listing 10). This function receives the same arguments as ENTMOD-Z, so they are interchangeable. In it we will use two of the five properties which are exclusive of MESH objects. These are the COORDINATE property and the VERTEXCOUNT property.
Table 19.4. AcadSubDMesh object properties.

	Property:
	Description:

	Coordinate
	Array of vertex coordinates. 

To get the 3D point for the first vertex: (vla-get-Coordinate obj 0) 

To set its value (vla-put-Coordinate obj 0 3DPoint). 

This expression throws an error if Smoothness is greater than 0.

	Coordinates
	Coordinates array. The X, Y and Z values appear one after the other, as elements of the safearray's same dimension. 

To get the number of vertices:

(/ (1+

    (vlax-safearray-get-u-bound

      (vlax-variant-value (vla-get-Coordinates obj))

       1))

   3)

	FaceCount
	Gets the number of faces in the MESH. Read only.

	Smoothness
	Gets or sets the smoothing level.

	VertexCount
	Gets the number of vertices.


The COORDINATE property returns an array of vertex coordinates. Each vertex coordinates can be retrieved using the VLA-GET-COORDINATE function using the zero-based index number that identifies each vertex. A new value can be given to a vertex’s coordinates using the VLA-PUT-COORDINATE function using the index number that identifies the vertex. To modify a vertex this way the MESH Smoothness, which is exposed as other of its properties, must be always ZERO. I would like to remark that this limitation is not present when using ENTMOD. ENTMOD can modify meshes with any smoothing level.
The VERTEXCOUNT property contains the number of vertices. It is a Read-only property which we will use to control a REPEAT loop. Its value can be retrieved by the VLA-GET-VERTEXCOUNT function.
We must initialize our environment when using the Visual LISP ActiveX interface. First of all, the Visual LISP ActiveX extensions must be expressly loaded using the VL-LOAD-COM function. The MESH ActiveX object must be retrieved from the ename received as argument using the VLAX-ENAME->VLA-OBJECT function. Finally the variable i, which will be used as the COORDINATE vertex index, is set to ZERO. 
The processing of our MESH’s vertices will be done in a REPEAT loop. The number of repetitions is the number of MESH vertices, which is retrieved from the VLA-OBJECT using the VLA-GET-VERTEXCOUNT function. Each vertex coordinate value retrieved with VLA-GET-COORDINATE using the index I is an array. From this array we extract the numerical values as a list using the function VLAX-SAFEARRAY->LIST. As this array is contained in a variant we must first extract it using VLAX-VARIANT-VALUE. The new Z value is calculated by applying the function contained in the EQUATION variable to the X and Y coordinate values and this new value is used to update the value of the COORDINATE property at the i index. The number list is transformed to the adequate format using the VLAX-3D-POINT function. To process the following vertex the I variable is incremented by 1 before repeating the loop.
When testing AX-MOD-Z with any of the EQUATION functions, remember that the function name must be preceded by an apostrophe symbol. 
(defun ax-mod-z  (ename equation / obj i pt)

  (vl-load-com)

  (setq obj (vlax-ename->vla-object ename)

        i   0)

  (repeat (vla-get-VertexCount obj)

    (setq

      pt (vlax-safearray->list

           (vlax-variant-value (vla-get-Coordinate obj i))))

    (vla-put-Coordinate

      obj

      i

      (vlax-3d-point

        (list (nth 0 pt)

              (nth 1 pt)

              (apply equation (list (nth 0 pt) (nth 1 pt))))))

    (setq i (1+ i))))

Listing 10. Modifying vertex Z coordinates through ActiveX (only WINDOWS).

The SHAPE-M command.
We have studied how we can modify the Z value as a function of the X and Y values. But we should be able to do so also for the X or Y values. This way we would have a really powerful MESH shaping command. The definition of this command, the SHAPE-M command is shown in Listing 11.

The SHAPE-M command includes a command line user interface that will prompt the user for the necessary information. Once this information is input, it will call the ENTMOD-XYZ function (Listing 11) which is a variation of the ENTMOD-Z function shown in Listing 9.

SHAPE-M begins by prompting for the MESH we want to shape up. Next it will prompt for the shaping direction offering X, Y or Z as options. Once we have selected the shaping direction, it will prompt for the shaping equation function’s name. This must be the name of a function that has been already loaded like the ones (S01 to S10) in the AU2012-MESH LISP code file included in the dataset for this class.

Finally it will prompt for the shaping offset. This is the target distance between the maximum and minimum calculated values. This value will be used for scaling the transformation in the selected direction. 

 (defun C:SHAPE-M  (/ ent axis equation offset)

  (command "_undo" "_begin")

  (prompt "\nSelect MESH to shape: ")

  (while (not (setq ent (ssget "_:S" '((0 . "MESH")))))

    (prompt "\nSelect MESH to shape: "))

  (initget (+ 2 4) "X Y Z")

  (setq axis

         (getkword "\nSpecify shaping axis [X/Y/Z]<Z>"))

  (cond ((= axis "X") (setq axis 0))

        ((= axis "Y") (setq axis 1))

        (t (setq axis 2)))

  (setq ent (ssname ent 0))

  (initget 1)

  (setq equation

         (getstring

           "\nSpecify shaping equation function: "))

  (cond

    ((car (atoms-family 0 (list equation)))

     (initget (+ 1 2 4))

     (setq offset (getdist "\nShaping offset: "))

     (entmod-xyz ent (read equation) offset axis))

    (t

     (prompt

       (strcat

         "\nFunction "

         equation

         " has not been defined in the current context."))))

  (command "_undo" "_end")

  (princ))
Listing 11. Mesh shaping program.
The ENTMOD-XYZ function 
The ENTMOD-XYZ function is the one that makes the vertex coordinates transformation. It will receive the four user specified values as its arguments. As EQUATION is entered as a string, it must be transformed into a symbol applying the READ function.
The AXIS argument defines the transformation direction. According to its value, the variables I and J are set to point to the other two coordinates:

(cond ((= axis 0) (setq i 1 j 2))

        ((= axis 1) (setq i 0 j 2))

        ((= axis 2) (setq i 0 j 1)))

 and the 3D point coordinate lists are ordered.
(cond ((= axis 0) (setq vertices-mod (cons (list v-mod v-i v-j) vertices-mod)))

          ((= axis 1) (setq vertices-mod (cons (list v-i v-mod v-j) vertices-mod)))

          ((= axis 2) (setq vertices-mod (cons (list v-i v-j v-mod) vertices-mod))))
The offset argument is used to calculate a scale factor to be applied to the computed coordinate values. To calculate this scale factor the values for all the vertices are computed in a MAPCAR expression, the minimum value is subtracted from the maximum and the OFFSET argument value is divided by this HEIGHT and assigned to the SCALE-F variable. The value computed for each vertex will later be multiplied by this scale factor.
  (setq lst-mod

         (mapcar

           '(lambda (vert) (apply equation (list (nth i vert) (nth j vert))))

           vertices)

        height  (- (apply 'max lst-mod) (apply 'min lst-mod))

        scale-f (/ offset height))

Finally, the MESH’s entity definition list will be modified in a WHILE loop just as we did in ENTMOD-Z and the entity is transformed in a call to ENTMOD using the modified entity list as argument.
  (setq k 0)

  (while (< k (length vertices))

    (setq ent-list

           (subst (cons 10 (nth k vertices-mod)) (cons 10 (nth k vertices)) ent-list)

          k        (1+ k)))

  (entmod ent-list)

The function concludes to a call to the ENTUPD function that updates the transformed entity so the modifications are immediately displayed. 

(defun entmod-XYZ  (ename equation offset axis / ent-list vertices i j k lst-mod

                    height scale-f v-i v-j v-mod vertices-mod)

  (setq ent-list (entget ename)

        vertices (values 10 ent-list))

  (cond ((= axis 0) (setq i 1 j 2))

        ((= axis 1) (setq i 0 j 2))

        ((= axis 2) (setq i 0 j 1)))

  (setq lst-mod

         (mapcar

           '(lambda (vert) (apply equation (list (nth i vert) (nth j vert))))

           vertices)

        height  (- (apply 'max lst-mod) (apply 'min lst-mod))

        scale-f (/ offset height))

  (foreach vertex  vertices

    (setq v-i   (nth i vertex)

          v-j   (nth j vertex)

          v-mod (+ (nth axis vertex) (* scale-f (apply equation (list v-i v-j)))))

    (cond ((= axis 0) (setq vertices-mod (cons (list v-mod v-i v-j) vertices-mod)))

          ((= axis 1) (setq vertices-mod (cons (list v-i v-mod v-j) vertices-mod)))

          ((= axis 2) (setq vertices-mod (cons (list v-i v-j v-mod) vertices-mod)))))

  (setq vertices-mod (reverse vertices-mod))

  (setq k 0)

  (while (< k (length vertices))

    (setq ent-list

           (subst (cons 10 (nth k vertices-mod)) (cons 10 (nth k vertices)) ent-list)

          k        (1+ k)))

  (entmod ent-list)

  (entupd ename))

Listing 12. Universal mesh vertex modifying function.
Example Surface Equation Functions.
We have included in AU2012-MESH.LSP ten functions that calculate a value as a function of two arguments I’ve called X and Y, but which can obviously be any numerical value. They are named as S01 to S10. The screen captures shown in our video presentation were made from a square 5 drawing units mesh with its center in the World Coordinate System’s origin. This mesh has 400 faces in 20 columns and 20 rows. The offset value used to shape them was three.
(defun s01 (x y) (* x y))

(defun s02 (x y /)

  (sqrt (abs (* x y))))

(defun s03 (x y)

  (+ (expt x 2) (expt y 2)))

(defun s04 (x y) ;Hyperbolic paraboloid

  (- (expt x 2) (expt y 2)))

(defun s05  (x y /) ;Revolution paraboloid

  (- (+ (expt x 2)

        (expt y 2))))

(defun s06  (x y)

  (- (+ (expt x 2) (expt y 2) (expt y 3)) (expt x 3)))

(defun s07 (x y /)

  (exp (- (+ (expt x 2) (expt y 2)))))

(defun s08  (x y /)

  (/ (sin (* pi x y)) (+ 0.5 (expt x 2) (exp 2))))

(defun s09  (x y /)

  (cos (* x y)))

(defun s10 (x y /)

  (sin (+ (expt x 2) (expt y 2))))

Summary.
The code I have used in this class has been simplified in order to make the essential concepts clear. This simplification has implied removing error handling which would be necessary for real-life applications. We have also resorted to using AutoCAD commands where in our usual programming style we would have been using ENTMAKE or ActiveX functions.

These finer aspects of AutoLISP/Visual LISP programming are covered in detail in my book AutoCAD expert’s Visual LISP, from which the contents of this class have been extracted. Should you be interested in the book, it is available at AMAZON.COM.
I hope the possibilities of AutoCAD’s improved 3D capabilities will be seen in a new light when understanding how they can also be the subject for customization through user programming. 
I would be pleased if I can help you in these tasks. You can reach me through my website, www.togores.net, through social networks like FaceBook or Linkedin or through my e-mail, reinaldo.togores@unican.es.

